

線形代数(II 逆行列)

—数値計算(08/11/7)—

関西学院大学理工学部 西谷滋人

Copyright ©2007, 08 by Shigeto R. Nishitani

行列計算

数値計算の中心課題の一つである、行列に関する演算について見ていく。多次元、大規模な行列に対する効率のよい計算法が多数開発されており、多くの既存のライブラリが用意されている。本章ではそれらの中心をなす、逆行列(matrix inverse)と固有値(Eigen values)に関して具体的な計算方法を示す。現実的な問題には既存のライブラリを使うのが上策であるが、それでも基礎となる原理の理解や、ちょっとした計算、ライブラリの結果の検証に使えるルーチンを示す。

逆行列は連立一次方程式を解くことと等価である。ルーチ的なやり方にガウスの消去法がある。これは上三角行列になれば代入を適宜おこなうことで解が容易に求まることを利用する。さらに、初期値から始めて次々に解に近づけていく反復法がある。この代表例であるJacobi(ヤコビ)法と、収束性を高めたGauss-Seidel(ガウス-ザイデル)法を紹介する。

上記の手法をより高速にした修正コレスキー分解と共役傾斜(共役勾配)法があるが、少し複雑になるので割愛する。必要ならばNumRecipeを読め。

ガウス消去法による連立一次方程式の解

原理と手作業

逆行列は連立一次方程式を解くことと等価である。すなわち、 A を行列、 x を未知数ベクトル、 b を数値ベクトルとすると、

$$\begin{aligned} Ax &= b \\ A^{-1}x &= A^{-1}b \\ x &= A^{-1}b \end{aligned}$$

である。未知数の少ない連立一次方程式では、適当に組み合わせで未知数を消していけばいいが、未知数が多くなってしまうと破綻する。未知数の多い多元連立一次方程式で、ルーチ的に解を求めていく方法がガウス消去法で、前進消去と後退代入という2つの操作からなる。

後退代入(Backward substitution)による解の求め方を先ず見よう。たとえば、

$$\begin{aligned} x+y-2z &= -4 \\ -3y+3z &= 9 \end{aligned}$$

$$-z = -2$$

では、下から順番に $z \rightarrow y \rightarrow x$ と適宜代入することによって、簡単に解を求めることが出来る。係数で作る行列でこのような形をした上三角行列にする操作を前進消去あるいはガウスの消去法(Gaussian elimination)という。下三角行列L(lower triangular matrix)と上三角行列U(upper triangular matrix)の積に分解する操作

$$A = LU$$

をLU分解(LU decomposition)という。例えば先に示した上三角行列を係数とする連立方程式は、

$$\begin{aligned} x+y-2z &= -4 \\ x-2y+z &= 5 \\ 2x-2y-z &= 2 \end{aligned}$$

を変形することで得られる。また、このL行列は

となる。

MapleによるLU分解

係数行列(coefficient matrix)と定数項(b)との関係は以下の通りである。

```
> restart;
A:=Matrix([[1,1,-2],[1,-2,1],[2,-2,-1]]);
X:=Vector([x,y,z]);
#X:=Vector([1,-1,2]);
b:=Vector([-4,5,2]);
A.X=b;
```

$$\begin{bmatrix} x+y-2z \\ x-2y+z \\ 2x-2y-z \end{bmatrix} = \begin{bmatrix} -4 \\ 5 \\ 2 \end{bmatrix}$$

単に逆行列を求める際は

```
> with(LinearAlgebra):
MatrixInverse(A);
```

(3.2.1)

$$\begin{bmatrix} \frac{4}{3} & \frac{5}{3} & -1 \\ 1 & 1 & -1 \\ \frac{2}{3} & \frac{4}{3} & -1 \end{bmatrix} \quad (3.2.1)$$

である。Mapleでは行列を三角行列に分解するために、LUdecompositionコマンドが用意されている。係数と定数項から作られる行列を拡大係数行列(augmented matrix)といい、 $\langle A|b \rangle$ とすると作られる。

```
> P,L,U:=LUdecomposition(<A|b>);
```

$$P, L, U := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 1 & 1 & 0 \\ 2 & \frac{4}{3} & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 & -2 & -4 \\ 0 & -3 & 3 & 9 \\ 0 & 0 & -1 & -2 \end{bmatrix}$$

後退代入までおこなってもとめた連立方程式の解は以下の通り。

```
> LUdecomposition(<A|b>, output='R');
```

$$\begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 2 \end{bmatrix} \quad (3.2.2)$$

▼ MapleによるLU分解のプログラム

LU分解すれば線形方程式の解が容易に求まることは理解できると思う。具体的にAをLU分解する行列(消去行列と称す)T1,T2の係数は次のようにして求められる。

```
> A0:=Matrix([[1,1,-2],[1,-2,1],[2,-2,-1]]):
b0:=Vector([-4,5,2]):
A:=Matrix(A0):
B:=Vector(b0):
n:=3:
L:=Matrix(array(1..n,1..n,identity)):
for i from 1 to n do
T[i]:=Matrix(array(1..n,1..n,identity)):
for j from i+1 to n do
am:=A[j,i]/A[i,i]:
T[i][j,i]:=-am:
L[j,i]:=am:
for k from 1 to n do
A[j,k]:=A[j,k]-am*A[i,k]:
end do:
B[j]:=B[j]-B[i]*am:
end do:
end do:
> T[1]:
T[2]:
```

$$\begin{bmatrix} 1 & 0 & 0 \\ -1 & 1 & 0 \\ -2 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & -\frac{4}{3} & 1 \end{bmatrix}$$

```
> U:=T[2].T[1].A0;
```

$$U := \begin{bmatrix} 1 & 1 & -2 \\ 0 & -3 & 3 \\ 0 & 0 & -1 \end{bmatrix}$$

```
> L.U;
```

$$\begin{bmatrix} 1 & 1 & -2 \\ 1 & -2 & 1 \\ 2 & -2 & -1 \end{bmatrix} \quad (3.3.1)$$

```
> L;A;
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 1 & 1 & 0 \\ 2 & \frac{4}{3} & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 1 & -2 \\ 0 & -3 & 3 \\ 0 & 0 & -1 \end{bmatrix} \quad (3.3.2)$$

```
> B;
```

$$\begin{bmatrix} -4 \\ 9 \\ -2 \end{bmatrix}$$

▼ ピボット操作

ガウス消去法で困ることに、もしある行を対角要素で割ろうとするときに、その対角要素が0になる場合がある。しかし、この場合にも、方程式の順序(行列の行と右辺の値)を入れ替えれば解決する。この割るほうの要素をピボット要素あるいはピボット(pivot)バスケの軸足を動かさずにくるくる回すやつと呼ぶ。この操作は変数の並びは変わらず単に方程式の順番を変更しただけである。

さらに対角要素の数値が厳密に0でなくとも極端に0に近づいた場合でも、その数で

割った数値が大きくなり他の数との差を取ると以前に示した情報落ちの可能性が出てくる。この現象を防ぐためには、絶対値が最大のピボットを選んで行の入れ替えを毎回おこなうといい結果が得られることが知られている。

MapleのLUDecompositionコマンドをこのような行列に適用すると、置換行列(permutation matrix)Pは単位行列ではなくなる。P.A=LUであることに注意。

反復法による連立方程式の解

以下のような連立方程式を

$$\begin{cases} 5x + y + z + u = -6 \\ x + 3y + z + u = 2 \\ x - 2y - 9z + u = -7 \\ x + 3y - 2z + 5u = 3 \end{cases}$$

形式的に解くと

$$x = \frac{-6 - (x + y + z)}{5}$$

y=

=

u=

となる。適当に初期値(x0,y0,z0,u0)をとり、下側の方程式に代入すると、得られた出力(x1,y1,z1,u1)はより正解に近い値となる。これを繰り返すことによって正解が得られる。これをヤコビ(Jacobi)法と呼び、係数行列の対角要素が非対角要素にくらべて大きいときに適用できる。多くの現実の問題ではこの状況が成り立っている。

Gauss-Seidel法はJacobi法の高速版である。n番目の解の組が得られた後に一度に次の解の組に入れ替えるのではなく、得られた解を順次改良した解として使っていく。これにより、収束が早まる。以下にはヤコビ法のコードを示した。x1[]の配列を変数に換えるだけで、Gauss-Seidel法となる。

```
> AA:=Matrix([[5,1,1,1],[1,3,1,1],[1,-2,-9,1],[1,3,-2,5]]):
b:=Vector([-6,2,-7,3]):
n:=4;
x0:=[0,0,0,0]:
```

```
x1:=[0,0,0,0]:
for iter from 1 to 20 do
for i from 1 to n do
x1[i]:=b[i];
for j from 1 to n do
x1[i]:=x1[i]-AA[i,j]*x0[j];
end do;
x1[i]:=x1[i]+AA[i,i]*x0[i];
x1[i]:=x1[i]/AA[i,i];
end do;
x0:=evalf(x1);
print(iter,x0);
end do;
```

n:=4

```
1, [-1.200000000, 0.666666667, 0.777777778, 0.600000000]
2, [-1.608888889, 0.607407407, 0.562962963, 0.751111112]
3, [-1.584296296, 0.764938271, 0.547489711, 0.782518518]
4, [-1.618989300, 0.751429355, 0.518705076, 0.676892181]
17, [-1.585703890, 0.828561399, 0.485749823, 0.614430394]
18, [-1.585748324, 0.828507889, 0.485733745, 0.614303868]
19, [-1.585709101, 0.828570237, 0.485726640, 0.614338429]
20, [-1.585727061, 0.828548010, 0.485720984, 0.614290334]
```

(4.1)

課題

1 後退代入法で解を求めよ。

$$\begin{cases} x+4y-3z=1 \\ -6y+4z=1 \\ -5/3z=1/3 \end{cases}$$

2 次の行列AをLU分解せよ。

```
> Matrix([[1,4,3],[1,-2,1],[2,-2,-1]]);
```

$$\begin{bmatrix} 1 & 4 & 3 \\ 1 & -2 & 1 \\ 2 & -2 & -1 \end{bmatrix}$$

(5.1)

3 Jacobi法のプログラムを参照してGauss-Seidel法のプログラムを作れ。Jacobi法と収束性を比べよ。

4 次の連立方程式の解を求めよ。ただし、pivot操作が必要となる。

```
> with(LinearAlgebra):
A:=Matrix([[3,2,2,1],[3,2,3,1],[1,-2,-3,1],[5,3,-2,5]]):
X:=Vector([w,x,y,z]):
b:=Vector([-6,2,-9,2]):
A.X=b;
```

(5.2)

$$\begin{cases} \begin{bmatrix} 3w+2x+2y+z \\ 3w+2x+3y+z \\ w-2x-3y+z \\ 5w+3x-2y+5z \end{bmatrix} = \begin{bmatrix} -6 \\ 2 \\ -9 \\ 2 \end{bmatrix} \end{cases} \quad (5.2)$$

5 pivot操作を含めたLU分解のプログラムを作成せよ。上の問題を解き、そのL, U行列および $L^{-1} \cdot b$ ベクトルを求めよ。